

Newton Software Exchange

INDEPENDENT
OSI USERS
NEWSLETTER

VOL. 1, NO. 3 MARCH, 1979

MAILING ADDRESS: P.O. Box 518, Newton Corner, MA 02158

GIVE YOUR OSI A RIDE ON THE S-100 BUS...

Glenn L. Buehl, 11 Dexter Street., Derry, NH 03038 sends us some interesting information for those of you trying to hook on the S-100 Bus. While he has not done this himself, he says that a request for info from: Forethought Products, Inc.; 87070 Dukhobar Rd.; Eugene OR. 97402. Phone (503) 485-8575 brings results.

This company builds KIMSI and BETSI S-100 bus converters. They support using the KIMSI with the OSI 48-line bus. They will send you an application note that covers the jumpering and interfacing required.

Anyone trying this should please keep us informed.

Glenn also says that an inquiry to SUB-LOGIC, BOX V, Savoy, IL. 61874 brought him a booklet showing several 3-D graphics routines, written in a universal Basic.

He has a few questions: He would like to add color, or change the 540 board from a 32x64 display to a 128x256 point plot mode. The 8-bit character cells are all pre-defined, he notes, and he wants to be able to make up his own character sets of plot points.

Check with Glenn if you are interested in buying a C2-4P with 8K RAM at a good price.

----- C2-4P MEMORY MAP REQUEST

Len Mangerman, Polaroid Corp., 565 Technology Square--5A, Cambridge, MA. 02139. Is expanding his C2-4P with a 502 board to 32K with a 527 board. He writes, "It would be a great help to everyone if you would publish a complete memory map of all addresses with a clear explanation of the function associated with them."

How about you dealers? Have any of you done this? If so we will be happy to let our readers know of availability.

Len would also know if anyone has patched the "Auto Load Cassette" program listed in the July '77 OSI Journal so that it works on the C2-P with the 502 board.

----- SCREEN MEMORY LOCATION FOR THE C2-4P

Irv Johnson, 1214 Ravine Street, Janesville, WI 5345 has created documentation to help find screen memory locations by the use of tables and a plotting chart. It is for screen memory D000-D7FF or 53248-55295. Send him \$1.50 for tables, chart and how to use info.

Irv has a C2-4P with 502 CPU, 540 board and polled keyboard. He wants to wire a 525 16K memory board (populated w 8K) and/or the CPU board to work in his computer.

```

10 REM POLLED KEYBOARD LUNAR LANDER
20 REM   By David Morgenstein
30 REM 9523 48th Place
40 REM College Park, MD 20740
100 PRINTCHR$(26):X=25:Y=10: GOSUB 6000
104 PRINT "L U N A R L A N D E R": Y=12: GOSUB 6000
106 INPUT"   Do You Need Instructions (Y/N) ";N$
110 IF N$="N" GOTO 190
115 PRINT:PRINT
120 PRINT TAB(10)"This is a real time Lunar Lander simulation.
130 PRINT TAB(10)"To play, merely enter the pounds of fuel which you
140 PRINT TAB(10)"wish to burn by typing a digit ( 0 - 9 ).
150 PRINT TAB(10)"The nine gives maximum burn, slowing you down at the
160 PRINT TAB(10)"fastest rate. A zero give no burn and lets you free
170 PRINT TAB(10)"fall."
180 PRINT : INPUT"   READY...TYPE GO   ";N$

190 PRINT CHR$(26): Y=4: X=28: GOSUB 6000: PRINT "Time to fuel exhaustion"
200 X=20: Y=7: GOSUB 6000: PRINT "BURN RATE"
220 X=50: GOSUB 6000: PRINT "FUEL"
225 REM trying to see if it is a memory problem
230 Y=8:X=20:GOSUB 6000:PRINT "(LBS/SEC)";:GOSUB 6000: X=50: PRINT"(LBS)"
240 Y=12:X=20: GOSUB 6000: PRINT"VELOCITY": X=50: GOSUB 6000:PRINT "ALTITUDE"
250 Y=13: X=20 : GOSUB 6000: PRINT"(FT/SEC)":X=50: GOSUB 6000: PRINT" (FT)"
260 Y=18: X=28: GOSUB 6000: PRINT "ESTIMATED TIME TO LANDING"
270 Y=22: X=1: GOSUB 6000: FOR I=1 TO 79: PRINT"-";: NEXT I
275 Y=23: X=1: GOSUB 6000: PRINT "0 "
280 FOR I= 1 TO 7: X=10*I: GOSUB 6000: PRINTI;: NEXT I
290 X=30: Y=24: GOSUB 6000: PRINT "ALTITUDE (X10,000 FT.)": GOSUB6000
310 VE=-100: MT$=" " : FU=10000: AL=80000: DE=5: BU=32

320 FOR T = 1 TO 10000
330 IF T/2=INT(T/2) THEN: PRINT CHR$(7)
340 VE=VE+((BU-32)*25E8)/((25E8+AL*AL))
345 VE=INT(VE)
350 AL=AL+INT(VE/2)
360 IF AL<0 GOTO 3000
370 IF FU<500 THEN: GOSUB 2000
380 FU=FU-BU/2
385 IF FU<=0 THEN FU=0: BU=0
390 IF BU<=0 THEN B$="NO BURN": GOTO 410
400 B$=STR$(INT(FU/BU))
410 X=38: Y=5: GOSUB 6000: PRINT MT$: GOSUB 6000: PRINT B$
420 X=21: Y=9: GOSUB 6000: PRINT BU: X=50: GOSUB 6000: PRINT FU
430 X=22: Y=14: GOSUB 6000: PRINT VE: X=50: GOSUB 6000: PRINT AL
440 IF VE>=0 THEN A$= "ESCAPE" : GOTO 460
450 A$=STR$(INT(AL/ABS(VE)))
460 Y=19: X=38: GOSUB 6000: PRINT MT$: GOSUB 6000: PRINTA$
465 TA=INT((AL+500)/1000): IF TA>80 THEN TA=80
470 IF TA<1 THEN TA=1
475 Y=21: X=TA+1: GOSUB 6000
480 IF FU=0 GOTO 500
485 GOSUB 5000: IF Z=13 GOTO 500
490 BU= 12+4*(Z-48)
495 IF Z=48 THEN BU=0
500 FOR TI= 1 TO DE: A=SIN(10): NEXT TI
505 VP=VE: AP=AL
510 NEXT T

```

```

2000 FOR J= 1 TO 2
2005 X=36: Y=12: GOSUB 6000: PRINT "LOW FUEL"
2010 Y=13: GOSUB 6000: PRINT "WARNING!"
2020 A=SIN(10)
2030 GOSUB 6000: PRINT MT$: Y=12: GOSUB 6000: PRINT MT$
2035 A=SIN(10)
2040 NEXT J
2050 DE=1
2060 RETURN

3000 SP=(VP+VE)/2
3010 IF SP<-25 GOTO 3200
3015 PRINT: PRINT
3020 PRINT TAB(20)"CONGRATULATIONS! You touched down at a mere"
3030 PRINT TAB(30)SP;" FT/SEC. A Safe Landing !!!"
3040 PRINT PRINTTAB(20)"Do you want to try again and"
3050 PRINT TAB(20)" ";: INPUT"prove it wasn't luck ";N$
3060 IF N$ = "N" THEN GOTO 520
3070 GOTO 190
3200 PRINT CHR$(26)
3210 N=40
3220 FOR I= 1 TO N: X=1+INT(79*RND(1)): Y=1+INT(23*RND(1))
3225 GOSUB 6000: PRINT CHR$(33+INT(15*RND(1))): GOSUB 6000: NEXT I
3230 X=20: Y=10: GOSUB 6000: PRINT "YOU JUST BLEW A CRATER,"
3240 Y=11: GOSUB 6000: PRINT ABS(VE);" FEET IN DIAMETER, ON THE"
3250 Y=12: GOSUB 6000: PRINT "SURFACE OF THE MOON. BETTER TRY AGAIN..."
3260 Y=14: GOSUB 6000: INPUT "READY (Y/N) ";N$
3270 GOTO 190

5000 Z=PEEK(64513)
5005 IF Z=13 THEN RETURN
5010 IF Z>128 THEN Z=Z-128: RETURN

6000 PRINT CHR$(27); CHR$(61);CHR$(Y+31);CHR$(X+31);: RETURN

```

LUNAR LANDER DEMONSTRATES HOW TO PEEK
YOUR WAY TO REAL TIME INTERACTION

The previous program by David Morgenstein includes a number of interesting concepts. The first is the way in which a serial terminal is polled in real time. A PEEK at 64513 (\$FC01) gives you the character sent by the terminal. See subroutine at 5000. It is necessary to mask the parity bit as indicated at 5010, and to test for, but skip 13, <CR>, as it is a null byte appearing before any keyboard entry has been made. The next important contribution, especially to those of you with an ADM3A, is the subroutine at 6000. This provides direct cursor positioning at X/Y coordinates, using a two-character ESC (CHR\$(27)) sequence. X value ranges

from 1 to 80; Y from 1 to 24. Line 500 is a delay line. DE can be changed in Line 310, and you might want to put it on a separate line just to see how it changes things. Your editor managed to Deep Space himself by not paying enough attention to velocity..

David says that timing changes when LOW FUEL WARNING comes on...maybe one of you can figure out why.

Thanks for the fun.

```

10 REM DOGFIGHT II
20 REM By Ken Willoughby
21 REM 4217 22nd St.
22 REM San Francisco, CA 94114
30 REM KEYS- LT:1,2,3 RT: 8,9,0
40 FOR N= 1 TO 26: PRINT : NEXT N
50 POKE 530,1
60 L=53736: Y=57088
70 R=53752
80 C=237
90 POKE Y, 191: POKE R,C
100 FOR D=1 TO 50: NEXT D

```

```

110 IF PEEK(Y)=127 THEN 160
120 IF PEEK(Y)=191 THEN 180
130 IF C=237 THEN 150
140 IF PEEK(Y)=223 THEN 200
150 GOTO 250
160 POKE R,32: R=R-32
170 POKE R,C: GOTO 250
180 POKE R,32: R=R+32
190 POKE R,C: GOTO 250

```

```

200 FOR F=R-1 TO R-20 STEP-1
210 POKE F,148
220 IF F=L THEN 410
230 FOR T=1 TO 15: NEXT T
240 POKE F,32: NEXT F
250 POKE Y,127: POKE L,C
260 IF PEEK(Y)=127 THEN 310
270 IF PEEK(Y)=191 THEN 330
280 IF C=239 THEN 300
290 IF PEEK(Y)=223 THEN 350
300 GOTO 480

```

```

310 POKE L,32: L=L-32
320 POKE L,C: GOTO 480
330 POKE L,32: L=L+32
340 POKE L,C: GOTO 480
350 FOR F=L+1 TO L+20
360 POKE F,148
370 IF F=R THEN 410
380 FOR T=1 TO 15: NEXT T
390 POKE F,32: NEXT F
400 GOTO 480

```

```

410 POKE F,32
420 PRINT TAB(10); "A HIT!"; CHR$(7)
430 INPUT B$: IF ASC(B$)>0 THEN 40
440 T=INT(RND(1)*50)
450 IF C=239 THEN 80
460 C=239
470 GOTO 90
480 T=T-0.5
490 IF T<0 THEN 440
500 GOTO 90

```

BATTLING IT OUT ON A CHALLENGER 1P
NO JOYSTICKS NEEDED FOR THIS ACTION

We are running the DOGFIGHT II listings on faith, as it does not run directly on our Challenger III. But I think Ken has worked it out and it certainly reads well enough. Let's hear from some of you 1P users to see how it runs.

This is a two player game. Each player control three keys (Left-1,2,3--Right-8,9,0). Key wants to add joysticks and a scoring subroutine.

If you delete line 100, the game runs at top speed.

VIDEO GRAPHICS WITH A 2P

Howard G. Drake
239 W. Main Street
Westboro, MA 01581

Sends us a program titled Drawing Board. It allows you to draw pictures using any of the graphics characters.

Howard has a 2P with graphics and a polled keyboard with BASIC-in-ROM and 8K RAM. He uses a Hitachi 12" BW TV with the Pickles and Trut conversion. "This gives me 29 lines by 64 characters that are very clear and legible," he says.

Howard will be happy to send a listing on cassette to those who send a blank cassette in a crush proof box and, of course, sufficient postage.

He also asks for a little help--
1) using Basic-in-ROM how can the cursor be positioned anywhere else on the screen except the lower left corner.
2) For a math program he needs 7-digit precision, but has only 6. Suggestions?
3) He would like to correspond with others interested in emulating a TI58/59 programmable calculator on a 2P.